


MS 365 Excel Basics 06: XLOOKUP, FILTER, LOOKUP, SWITCH Functions to Master Any Lookup Task!

Table of Contents

Important LOOKUP Functions	2
You must know Excel well before using AI.....	4
Example of AI provided formulas and then better formulas that are not provided:	5
XLOOKUP function arguments.....	6
No duplicates in lookup_array argument.....	7
Duplicates in lookup_array argument.....	8
Exact match lookup	9
Exact match lookup to a lookup product price.....	9
Data Validation dropdown list.....	10
Lookup an employee record	11
Lookup a column of day's sales and then add	11
Exact match or next smaller lookup	12
Lookup a tax rate: XLOOKUP or LOOKUP function?.....	13
LOOKUP function arguments.....	14
Why the Binary Search in the LOOKUP function is fast	15
Timing Results for XLOOKUP and LOOKUP functions	16
How XLOOKUP Binary may yield different answers than LOOKUP Binary when there are duplicates?.....	17
How can XLOOKUP yield same answer as LOOKUP, even with duplicate values?	18
Why use LOOKUP function rather than XLOOKUP function?.....	19
XLOOKUP can do vertical or horizontal lookup	19
XLOOKUP to perform Exact Match or Next Bigger lookup: lookup pipe size	19
XLOOKUP with wild card lookup: "Coca Cola" matches "Coca Cola Inc."	20
Lookup first when there are duplicates: lookup first day worked.....	21
Lookup last when there are duplicates: lookup last sale for employee.....	21
Two lookup values: lookup product price for a specified city	22
Two-Way lookup: lookup tax based on weekly pay and allowances.....	23
FILTER array function arguments	24
FILTER function to return multiple items from one lookup value	26
XMATCH function arguments.....	27
XMATCH lookup function to compare two lists	27
SWITCH function arguments	27
SWITCH to lookup the correct pricing lookup table based on product name	28
LOOKUP function to create formula much more quickly for tax calculations.....	29

Important LOOKUP Functions

XLOOKUP lookup a value from a cell Lookup a column Lookup a row Perform exact match lookup Perform exact match or next smaller Perform exact match or next bigger Perform wildcard lookup Perform binary search Search first-to-last Search last-to-first Lookup a cell reference Lookup a range & much more!		FILTER One lookup value, return multiple items
		XMATCH Return relative position of an item in a list Compare Two Lists
		LOOKUP Binary Search - REALLY fast Use when lookup table first column sorted A-Z like all tax tables, commission tables & rate tables Benefit is that formula is easier and faster to create
		SWITCH Can lookup anything: formulas, tables,

Not Covered in Video, but important too:

INDEX

Uses row and column numbers to lookup values from an array

CHOOSE

Uses sequential numbers to lookup items from within the arguments of CHOOSE


You must know Excel well before using AI

When you ask AI a question about building a solution in Excel, AI WILL give you an answer, but the answer may not always be a very good answer.

For example, if you ask a reasonable question like this:

I would like an Excel formula that can assign an "A+" grade to 95% or higher, an "A" grade to less than 95% and greater than or equal to 90%, a "B+" grade to less than 90% and greater than or equal to 85%, a "B" grade to less than 85% and greater than or equal to 80%, a "C+" grade to less than 80% and greater than or equal to 75%, a "C" grade to less than 75% and greater than or equal to 70%, a "D+" grade to less than 70% and greater than or equal to 65%, a "D" grade to less than 65% and greater than or equal to 60%, and a "F" grade to all other scores that are less than 60%.

This is what Chat GPT 4 returns as an answer:

 To create a formula that assigns grades based on percentages, you can use Excel's `IFS` function (available in newer versions) or a nested `IF` formula. Here's how you can write this formula in Excel:

Formula using `IFS` :

Assuming the percentage is in cell `A1`, the formula would be:

```
excel Copy code
```

```
=IFS(  
  A1>=95%, "A+",  
  A1>=90%, "A",  
  A1>=85%, "B+",  
  A1>=80%, "B",  
  A1>=75%, "C+",  
  A1>=70%, "C",  
  A1>=65%, "D+",  
  A1>=60%, "D",  
  TRUE, "F"  
)
```

Formula using nested `IF` :

If you're using an older version of Excel, use this formula:

```
excel Copy code
```

```
=IF(A1>=95%, "A+",  
  IF(A1>=90%, "A",  
    IF(A1>=85%, "B+",  
      IF(A1>=80%, "B",  
        IF(A1>=75%, "C+",  
          IF(A1>=70%, "C",  
            IF(A1>=65%, "D+",  
              IF(A1>=60%, "D", "F"))))))))
```

Replace `A1` with the actual cell containing the score for each student.

Let me know if you'd like additional help applying this!

The problems with this are:

1. Too Complicated.
2. % Values must be evaluated inside formula to get the decimal amount.
3. Violates Excel's Golden Rule: If a formula input can change, put it into a cell, label it and refer to it in the formula with a cell reference.


This means you must learn Excel well if you want to use AI to find Excel solutions – this way, when AI gives you a bad answer, you can decipher that and not use the answer.

Example of AI provided formulas and then better formulas that are not provided:

1	A	B	C	D	E	F	G	H	I	J	K	L
2	When you ask AI a question about building a solution in Excel, AI WILL give you an answer, but the answer may not always be a very good answer.											
3	For example, if you ask a reasonable question like this:											
4	I would like an Excel formula that can assign an "A+" grade to 95% or higher, an "A" grade to less than 95% and greater than or equal to 90%, a "B+" grade to less than 90% and greater than or equal to 85%, a "B" grade to less than 85% and greater than or equal to 80%, a "C+" grade to less than 80% and greater than or equal to 75%, a "C" grade to less than 75% and greater than or equal to 70%, a "D+" grade to less than 70% and greater than or equal to 65%, a "D" grade to less than 65% and greater than or equal to 60%, and a "F" grade to all other scores that are less than 60%.											
5												
6												
7												
8												
9	This is what Chat GPT 4 returns as an answer:											
10	D20: =IFS(C20>=95%, "A+", C20>=90%, "A", C20>=85%, "B+", C20>=80%, "B", C20>=75%, "C+", C20>=70%, "C", C20>=65%, "D+", C20>=60%, "D", TRUE, "F")											
11	E20: =IF(C20>=95%, "A+",IF(C20>=90%, "A",IF(C20>=85%, "B+",IF(C20>=80%, "B",IF(C20>=75%, "C+",IF(C20>=70%, "C",IF(C20>=65%, "D+",IF(C20>=60%, "D", "F"))))))))											
12												
13	Problems:											
14	Too Complicated.											
15	% Values must be evaluated											
16	Violates Excel's Golden Rule											
17												
18												
19												
20												
21												
22												
23												
24												
25												
26												
27												
28												
29												
30												
31												
32												
33												

XLOOKUP function arguments

The XLOOKUP function was introduced in 2019 and has replaced older lookup functions such as VLOOKUP, HLOOKUP and INDEX. The XLOOKUP function can lookup a cell, a column, a row, a cell reference, a range, and it can also perform various types of lookup such as: exact match, exact match or next smaller, exact match or next bigger, wildcard lookup, binary search, search first-to-last, search last-to-first and other types of lookups. There are a few other lookup functions that we still use because they can do things that XLOOKUP cannot do, or do things more efficiently than XLOOKUP can, such as the lookup functions: FILTER, XMATCH, LOOKUP, and SWITCH. We will see examples of each of these lookup functions also.



No duplicates in lookup_array argument

When there are no duplicates in the lookup_array argument , the order of records and search direction has no effect on returned value:


	A	B	C	D	E	F	G	H	I
2	When there are no duplicates, order of records and search direction has no effect on returned value:								
4	Search First To Last				Search Last To First				
5	Exact Match				Exact Match				
6	C8: =XLOOKUP(C7,B11:B14,C11:C14)				H8: =XLOOKUP(H7,G11:G14,H11:H14,,,-1)				
7	Product		Carlota		Product		Carlota		
8	Price		32		Price		32		
9									
10	Product		Price		Product		Price		
11	Quad		43		Quad		43		
12	Carlota		32		Carlota		32		
13	Sunshine		28		Sunshine		28		
14	Aspen		25		Aspen		25		
15									
16	Exact Match or Next Smaller				Exact Match or Next Smaller				
17	C19: =XLOOKUP(C18,B22:B25,C22:C25,,,-1)				H19: =XLOOKUP(H18,G22:G25,H22:H25,,,-1,-1)				
18	Sales		1250		Sales		1250		
19	Price		0.2		Price		0.2		
20									
21	Sales		Discount		Sales		Price		
22	0		0		0		0		
23	2500		0.4		1000		0.2		
24	1000		0.2		500		0.1		
25	500		0.1		2500		0.4		
26									
27	Exact Match or Next Larger				Exact Match or Next Larger				
28	C30: =XLOOKUP(C29,B33:B36,C33:C36,,1)				H30: =XLOOKUP(H29,G33:G36,H33:H36,,1,-1)				
29	Sales		1250		Sales		1250		
30	Price		0.4		Price		0.4		
31									
32	Sales		Discount		Sales		Price		
33	0		0		0		0		
34	2500		0.4		1000		0.2		
35	1000		0.2		500		0.1		
36	500		0.1		2500		0.4		

Duplicates in lookup_array argument

When there are duplicates, the order of records and search direction can affect the returned value:

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
2	When there are duplicates, the order of records and search direction can effect the returned value:													
4	Search First To Last				Search First To Last						Search Last To First			
5	Goal: Get first sale for SalesRep				Goal: Get first sale for SalesRep						Goal: Get last sale for SalesRep			
6	Exact Match				Exact Match						Exact Match			
7	C9: =XLOOKUP(C8,C12:C17,D12:D17)				G9: =XLOOKUP(G8,G12:G17,H12:H17)						L9: =XLOOKUP(L8,L12:L17,M12:M17,,,-1)			
8	SalesRep		Sioux		SalesRep		Sioux		Order of records in table changes, so result is different		SalesRep		Sioux	
9	Sales		640.56		Sales		1589.69				Sales		474.26	
10														
11	Date			SalesRep		Sales		Date			SalesRep		Sales	
12	10/3/2021			Sioux		\$640.56		10/3/2021			Chin		\$1,706.02	
13	10/3/2021			Chin		\$1,706.02		10/4/2021			Sioux		\$1,589.69	
14	10/4/2021			Sioux		\$1,589.69		10/5/2021			Chin		\$786.13	
15	10/4/2021			Chin		\$695.31		10/4/2021			Chin		\$695.31	
16	10/5/2021			Sioux		\$474.26		10/3/2021			Sioux		\$640.56	
17	10/5/2021			Chin		\$786.13		10/5/2021			Sioux		\$474.26	
18	Note: The match_modes <i>Exact match or next smaller</i> and <i>Exact match or next larger</i> behave the same as the Exact Match example above.													

Exact match lookup



Exact match lookup to a lookup product price

	A	B	C	D	E	F	G	H	I
1									
2		Goal: Lookup product price using "Exact match" lookup							
3									
4		Product	Price		Formula in C5:				
5		Quad	43.95		=XLOOKUP(B5,Product[Product],Product[Price])				
6									
7		Lookup Table:							
8									
9		ProductID	Product	Cost	Price				
10		A25C-4884	Aspen	14.55	26.95				
11		C20G-4398	Carlota	11.18	27.95				
12		Q22G-9634	Quad	19.34	43.95				
13		S35G-4382	Sunshine	11.25	22.95				
14		Y30C-4942	Yanaki	12.88	29.95				

Data Validation dropdown list

When you have a lookup table and you are doing Exact Match Lookup, you can use the unique list from the first column of the lookup table to help add a dropdown list of values to the cell that contains the lookup value for the XLOOKUP function. Here are the steps:

1. Select cell that contains the lookup value
2. Click the Data tab, then in the Data Tools group click the Data Validation button

The screenshot shows the Excel ribbon with the 'Data' tab selected. The 'Data Tools' group contains the 'Data Validation' button, which is highlighted by a black arrow. A tooltip for the 'Data Validation' button is visible, explaining its purpose: 'Pick from a list of rules to limit the type of data that can be entered in a cell. For example, you can provide a list of values, like 1, 2, and 3, or only allow numbers greater than 1000 as valid entries.' Below the ribbon, the spreadsheet shows a goal: 'Lookup product price using "Exact match" lookup'. A table with columns 'Product' and 'Price' is shown, with 'Quad' in cell B5. A formula in C5 is indicated as '=XLOOKUP(B5,Product[Product],Product[Price])'. A black arrow points to the 'Data Validation' button in the ribbon.

3. In the Data Validation dialog box, select "List" from the Allow textbox dropdown
4. Click in the Source textbox and select the unique list from the first column of the lookup table.
5. Click OK

The screenshot shows the Excel spreadsheet with the 'Data Validation' dialog box open. The 'Settings' tab is selected. The 'Validation criteria' section shows 'Allow:' set to 'List' and 'Source:' set to '=C510:C514'. The 'Data:' section is set to 'between'. The 'Ignore blank' and 'In-cell dropdown' checkboxes are checked. A black arrow points to the 'List' dropdown in the 'Allow:' section. Another black arrow points to the 'Source:' text box, which contains the formula '=C510:C514'. Below the dialog box, the spreadsheet shows the 'Lookup Table' with columns 'ProductID', 'Product', 'Cost', and 'Price'. The 'Product' column contains the values 'Aspen', 'Carlota', 'Quad', 'Sunshine', and 'Yanaki'. The 'Price' column contains the values '14.55', '27.95', '43.95', '22.95', and '29.95'. A black arrow points to the 'Quad' cell in the 'Product' column of the lookup table. The formula in C5 is shown as '=XLOOKUP(B5,Product[Product],Product[Price])'. The 'Data Validation' dialog box is positioned over the spreadsheet, with a black arrow pointing to the 'List' dropdown in the 'Allow:' section.

Lookup an employee record


	A	B	C	D	E	F	G	H
1								
2		Goal: Lookup employee record based on EmployeeID in cell B5.						
3								
4		EmployeeID		First	Last	StartDate	CellPhone	Department
5		4369-4774		Ty	Smithe	40862	435-398-5510	Finance
6								
7		Formula in D5:	=XLOOKUP(B5,B10:B13,C10:G13)					
8								
9		EmployeeID	First	Last	StartDate	CellPhone	Department	
10		4369-9084	Sioux	Chin	10/6/12	206-767-2190	Accounting	
11		4369-4774	Ty	Smithe	11/15/11	435-398-5510	Finance	
12		4369-2234	Gigi	Sy	2/3/18	206-337-0288	Accounting	
13		4369-3979	Kip	Hensel	10/6/20	206-821-4452	Maintenance	

Lookup a column of day's sales and then add

	A	B	C	D	E	F	G	H
1								
2		Goal: Lookup column and then aggregate as sum.						
3		Add total units by Day selected in cell C5.						
4								
5		Select Day:	Day 3		Formula in C6:			
6		Total Units	10,849		=SUM(XLOOKUP(C5,C10:F10,C11:F22))			
7								
8		Units by Day:						
9								
10		Hour	Day 1	Day 2	Day 3	Day 4		
11		6:00 AM	373	1,646	1,606	738		
12		7:00 AM	38	1,198	1,719	1,888		
13		8:00 AM	300	1,828	1,285	816		
14		9:00 AM	1,221	265	484	1,277		
15		10:00 AM	930	780	246	15		
16		11:00 AM	914	227	409	443		
17		12:00 PM	1,271	945	1,924	715		
18		1:00 PM	1,840	617	144	887		
19		2:00 PM	966	1,824	268	378		
20		3:00 PM	672	1,208	1,946	528		
21		4:00 PM	1,177	822	323	760		
22		5:00 PM	304	432	495	364		

Exact match or next smaller lookup

Goal: Lookup earnings tax rate using "Exact match or next smaller lookup"



Note: Tax, commission, discount and other similar lookup tables almost always have the first column sorted A-Z and almost never contains no duplicates.


Lookup a tax rate: XLOOKUP or LOOKUP function?

A	B	C	D	E	F	G	H	I	J	K
2	Goal: Lookup earnings tax rate using "Exact match or next smaller lookup"									
4	Earnings	TaxRate: XLOOKUP	Formula in C5:							
5	\$1,000	10.0%	=XLOOKUP(B5,B19:B24,D19:D24,0,-1)	< Much Slower calculating, duplicates may yield different answers						
7	Earnings	TaxRate: XLOOKUP (Binary)	Formula in C8:							
8	\$1,000	10.0%	=XLOOKUP(B8,B19:B24,D19:D24,0,-1,2)	< Slower calculating, duplicates may yield different answers						
10	Earnings	TaxRate: XLOOKUP	Formula in C11:							
11	\$1,000	10.0%	=XLOOKUP(B11,B19:B24,D19:D24,0,-1,-1)	< Much slower calculating, if duplicates will yield same results as LOOKUP						
13	Earnings	TaxRate: LOOKUP (Binary Search)	Formula in C14:							
14	\$1,000	10.0%	=LOOKUP(B14,B19:D24)	< Easier to create & faster calculation time						
16	Lookup Table:									
18	Earnings	Range	TaxRate							
19	\$0	\$0 - \$299.99	0.0%							
20	\$300	\$300 - \$599.99	2.5%							
21	\$600	\$600 - \$999.99	7.0%							
22	\$1,000	\$1,000 - \$1,999.99	10.0%							
23	\$2,000	\$2,000 - \$4,999.99	15.0%							
24	\$5,000	\$5,000 or more	20.0%							
26	Earnings	TaxRate	Formula in C27:							
27	-\$100	Negative Numbers Not Allowed	=XLOOKUP(B27,B19:B24,D19:D24,"Negative Numbers Not Allowed",-1)							
29	Earnings	TaxRate	Formula in C30:							
30	-\$100	0.0%	=XLOOKUP(B30,B19:B24,D19:D24,0,-1)							

LOOKUP function arguments

The LOOKUP is the first lookup function in spreadsheet history. The LOOKUP function was one of the original 21 functions in the first spreadsheet in history introduced in 1979: VisiCalc.

The first thing to note about the LOOKUP function is that it has two options—in essence giving you two different functions in one function (see Figure below):



The second argument of LOOKUP, **lookup_vector** or **array**, is the trigger for the function to determine whether to run Option 1 or Option 2. If you put a one-way array (column or row, but not table) into the second argument, Option 1 is executed; if you put a table (rows and columns) into the second argument, Option 2 is executed.

Consider the following information about Option 1, **LOOKUP(lookup_value, lookup_vector, [result_vector])**:

- This option searches for the lookup value in the ascending-sorted **lookup_vector** to find a match and yield a relative position that it uses to then retrieve the corresponding element in the argument **[result_vector]**.
- The **lookup_vector** must be sorted ascending, A–Z.
- The **lookup_vector** and **[return_vector]** argument values must be one-way arrays. These one-way arrays do not have to have the same orientation or the same number of elements. This is different from how the second and third arguments in XLOOKUP work.

Consider the following information about Option 2, **LOOKUP(lookup_value, array)**:

- This option allows the user to search for the lookup_value in the first column of the lookup table (array argument) and then return an item in the same corresponding position from the last column of the lookup table (array argument).

- The array argument allows you to either use a vertical lookup table or a horizontal lookup table.
 - If the number of rows in the lookup table is greater than or equal to the number of columns, LOOKUP does a vertical lookup. With a vertical lookup, the first column of the table must be sorted ascending, A–Z. The LOOKUP function tries to find a match and yields a relative position that it then uses to retrieve the corresponding element from the last column in the vertical lookup table.
 - If the number of columns in the lookup table is greater than the number of rows, LOOKUP does a horizontal lookup. With a horizontal lookup, the first row of the table must be sorted ascending, A–Z. The LOOKUP function tries to find a match and yields a relative position that it then uses to retrieve the corresponding element from the last row in the horizontal lookup table.
- The advantage that this option has over the XLOOKUP function is that if you have a table that is sorted correctly (like most tax tables, commission tables, and other tables), you only have to enter two arguments when using the LOOKUP function rather than enter five with the XLOOKUP function.

Why the Binary Search in the LOOKUP function is fast

Binary Search (Approximate Match) is faster than Linear Search (Exact Match)
 Binary Search reduces search time because it repeatedly divides the table in half and checks the one in the middle to help reduce calc time. It doesn't have to check each one.

For LOOKUP function, here is the idea:

Lookup	Returned Vale	
27	Above Ave	=LOOKUP(C8,C11:D16)

Lookup	Returned Vale	
0	Poor	
15	Below Ave	
20	Okay	
22	Ave	27>=22 ==> TRUE
24	Above Ave	takes upper half
29	Good	and

If this was false, it would take the lower half

22	Ave	divides in half again
24	Above Ave	27>=24 ==> TRUE
29	Good	takes upper half
		and

24	Above Ave	divides in half again
29	Good	27>=29 ==> FALSE

24	Above Ave	==>> Selects "Above Ave"
----	-----------	--------------------------

Timing Results for XLOOKUP and LOOKUP functions

	A	B	C	D	E	F	G	H	I
2							Times (mill sec)	Times (mill sec)	Times (mill sec)
3							219	22350	371
4							214	23337	373
5							215	29591	372
6							216	25093	372
7	Commission							11,517.0% increase	72.2% increase
8	Lookup Table			Sales Table					
9	2034 rows			756,442 rows					
10								Always yields same results as LOOKUP, but VERY SLOW	XL Binary, but slower than LOOKUP, not always same result as LOOKUP
11	Sales	Commission	Date	Sale	Binary Search LOOKUP	Exact Match or Next Smaller & Last-To-First XL(,,,,-1,-1)	Exact Match or Next Smaller & Binary Search XL(,,,,-1,2)		
12	0	0.0000%	7/25/24	406	0.000080	0.000080	0.000080		
13	50	0.0010%	8/8/25	510	0.000100	0.000100	0.000100		
14	100	0.0020%	12/6/24	33,049	0.006600	0.006600	0.006600		
15	150	0.0030%	6/18/25	4,757	0.000950	0.000950	0.000950		
16	200	0.0040%	1/15/25	155	0.000030	0.000030	0.000030		
17	250	0.0050%	5/27/24	4,215	0.000840	0.000840	0.000840		
18	300	0.0060%	5/12/24	394	0.000070	0.000070	0.000070		
19	350	0.0070%	10/12/25	311	0.000060	0.000060	0.000060		
2044	101,600	2.0320%	6/3/24	486	0.000090	0.000090	0.000090		
2045	101,650	2.0330%	1/8/25	408	0.000080	0.000080	0.000080		
2046			7/24/24	56	0.000010	0.000010	0.000010		
756451			10/2/25	20,166	0.004030	0.004030	0.004030		
756452			8/3/24	578	0.000110	0.000110	0.000110		
756453			9/26/24	108,959,725	0.020330	0.020330	0.020330		

How XLOOKUP Binary may yield different answers than LOOKUP Binary when there are duplicates?

LOOKUP & Binary Search calculations steps

Lookup Value: 2
 LOOKUP(lv, lookup_table) 5 =LOOKUP(F5,C10:D15)

Lookup Table:

1	4
1	6
2	10
2	5
3	16
3	22

Step 1

1	4
1	6
2	10

Checks if 2 (lv) >= 2 (la) and finds that this is TRUE
 And so it takes the upper values (table on bottom)

2	5
3	16
3	22

Step 2

2	5
---	---

Checks if 2 (lv) >= 3 (la) and finds that this is FALSE
 And so it takes the lower values (table on top)

2	5
---	---

Step 3

2	5
---	---

This is one row, so LOOKUP returns 5

XLOOKUP & Exact Match or Next Smaller & Binary Search calculations steps

(XLOOKUP Binary Search, but not same results as LOOKUP Binary Search and the calculation time is slower than the Binary)

Lookup Value: 2
 XLOOKUP(lv,la,ra,,-1,2) 10 =XLOOKUP(Z5,W10:W15,X10:X15,, -1,2)

Lookup Table:

1	4
1	6
2	10
2	5
3	16
3	22

Step 1

1	4
1	6
2	10

Checks if 2 (lv) <= 2 (la) and finds that this is TRUE
 And so it takes the lower values (table at top)

2	5
3	16
3	22

Step 2

1	4
---	---

Checks if 2 (lv) <= 1 (la) and finds that this is FALSE
 And so it takes the upper values (bottom at top)

1	6
2	10

Step 3

1	4
---	---

Checks if 2 (lv) <= 2 (la) and finds that this is TRUE and because this is last row, XLOOKUP returns 10

2	10
---	----

How can XLOOKUP yield same answer as LOOKUP, even with duplicate values?

XLOOKUP & Exact Match or Next Smaller & Search Last To First calculations steps

(Same results as LOOKUP Binary Search, but not a Binary Search and the calculation time is slower than the Binary)

Lookup Value: 2
 XLOOKUP(lv,la,ra,,-1,-1) 5 =XLOOKUP(O5,L13:L18,M13:M18,,-1,-1)

Because this is not a Binary Search that requires a sorted lookup_array, the XLOOKUP array-formula-type-algorithm works something like this:

Lookup Table:

1	4
1	6
2	10
3	16
2	5
3	22

Step 1: lv >= lookup_array

TRUE	=O5>=L13:L18
TRUE	
TRUE	
FALSE	
TRUE	
FALSE	

Step 2: Position of TRUE values?

1	=IF(O13#,SEQUENCE(ROWS(L13:L18)))
2	
3	
FALSE	
5	
FALSE	

Step 3:

If the search_mode is Last-to-first, then the max position is used to retrieve a value from the return_array:

Returned Value: 5 =INDEX(M13:M18,MAX(R13#))

If the search_mode is First-to-last, then the min position is used to retrieve a value from the return_array:

Returned Value: 4 =INDEX(M13:M18,MIN(R13#))

LOOKUP & Binary Search calculations steps

Lookup Value: 2
 LOOKUP(lv, lookup_table) 5 =LOOKUP(F5,C10:D15)

Lookup Table:

1	4
1	6
2	10
2	5
3	16
3	22

Step 1

1	4
1	6
2	10

2	5
3	16
3	22

Checks if 2 (lv) >= 2 (la) and finds that this is TRUE
 And so it takes the upper values (table on bottom)

Step 2

2	5
---	---

3	16
3	22

Checks if 2 (lv) >= 3 (la) and finds that this is FALSE
 And so it takes the lower values (table on top)

Step 3

2	5
---	---

This is one row, so LOOKUP returns 5

When to use LOOKUP function rather than XLOOKUP function?

If you need to do “Exact Match or Next Smaller” lookup, also known as “Approximate Match Lookup” and you are using a typical tax, commission, sales discount, or other similar tables that have the first column sorted A-Z, the first column has no duplicates and the last column has the values you want to return, use the LOOKUP function rather than the XLOOKUP function because of these benefits:

1. It takes less time to create the lookup formula with the LOOKUP function because you only have to enter 2 arguments rather than five or six.
2. The LOOKUP function uses a binary search process that allows the formula to calculate faster than the XLOOKUP function with a binary search option selected.

XLOOKUP can do vertical or horizontal lookup

	A	B	C	D	E	F	G	H	I	J
2		Goal: Lookup product price.								
4		Product	Carlota		Product	Carlota				
5		Price	27.95		Price	27.95				
7		Formula in C5: =XLOOKUP(C4,B12:B16,C12:C16)				Formula in F5: =XLOOKUP(F4,F11:J11,F12:J12)				
9		XLOOKUP can use vertical or horizontal lookup tables:								
11		Product	Price		Product	Aspen	Carlota	Quad	Sunshine	Yanaki
12		Aspen	26.95		Price	26.95	27.95	43.95	22.95	29.95
13		Carlota	27.95							
14		Quad	43.95							
15		Sunshine	22.95							
16		Yanaki	29.95							

XLOOKUP to perform Exact Match or Next Bigger lookup: lookup pipe size

	A	B	C	D	E	F	G	H	
2		Goal: Retrieve land square footage based on pipe size.							
4		Land Square Footage	23,000		Formula in C5:				
5		Pipe Size Required	8"		=XLOOKUP(C4,B8:B14,C8:C14,"Enter value "&B8&" or smaller",1)				
7		Max Land Square Footage for Pipe Size	Drain Pipe Size (in.)		Implied category:				
8		238,000	16"		238,000 >= Pipe Size > 134,000				
9		134,000	12"		134,000 >= Pipe Size > 83,000				
10		83,000	10"		83,000 >= Pipe Size > 46,000				
11		46,000	8"		46,000 >= Pipe Size > 22,000				
12		22,000	6"		22,000 >= Pipe Size > 14,000				
13		14,000	5"		14,000 >= Pipe Size > 3,500				
14		3,500	3"		3,500 >= Pipe Size > 0				

XLOOKUP with wild card lookup: “Coca Cola” matches “Coca Cola Inc.”

A few special characters can help you conduct searches in an Excel worksheet:

Asterisk (*): The * is a wildcard character that represents any set of zero or more characters.

Question mark (?): The ? is a wildcard character that represents exactly any one single character.

Tilde (~): The ~ is used to convert the wildcards * and ? to actual characters.

The figure below shows a lookup situation where you use a 2 in the *[match_mode]* argument so you can use wildcards with the lookup value. The goal is to type just part of the company name, such as *Coca Cola*, and get a match with the full company name *Coca Cola Inc.* The wildcard asterisk (*) represents zero or more characters. When you use the join operator (&) to join an asterisk in double quotes to the front and back of the cell reference B6, you are asking to make a match with any text in the *lookup_array* argument that contains *Coca Cola*. The formula also includes the text “Not Found” in the *[if_not_found]* argument so that XLOOKUP can return that message if the user types some text that is not contained in any of the values in the *lookup_array* argument.

	A	B	C	D	E	F	G	H	I
1									
2		Goal: Lookup Coca Cola and get a match for Coca Cola Inc.							
3		Example of: Wildcard lookup (synonyms: partial text lookup, fuzzy lookup)							
4									
5		Company (partial text)	City		Formula in C6:				
6		Coca Cola	Atlanta		=XLOOKUP("*"&B6&"*",B11:B14,C11:C14,"Not Found",2)				
7									
8		Lookup Table:							
9									
10		Company (full text)	City						
11		Coca Cola Inc.	Atlanta						
12		Pepsi Cola Inc.	NY						
13		RC Cola	KC						
14		Shasta Drinks	Calistoga						

The drawback to using wildcards is that it is not guaranteed that the formula can always return the correct value. For example, if you type Cola into cell B6, the formula will not know whether you want Coca Cola Inc., Pepsi Cola Inc., or RC Cola. However, because the default is to return the first item when there are duplicates, if you enter Cola, the formula will return the city of Atlanta because Coca Cola Inc. is the first company name that contains Cola. This sort of lookup is never 100% guaranteed to work.

Lookup first when there are duplicates: lookup first day worked

	A	B	C	D	E	F	G	H
1								
2		Goal: Lookup first date worked on project.						
3		"Get First" when there are duplicates = Exact Match Lookup does this by default.						
4								
5		"x" marks when employee worked on project.						
6								
7		Employee	First Date Worked	10/4/2021	10/5/2021	10/6/2021	10/7/2021	10/8/2021
8		Chin	10/6/2021			x	x	x
9		Gigi	10/5/2021		x		x	
10		Sioux	10/5/2021		x	x	x	
11		Chantel	10/4/2021	x	x	x	x	x
12		Billy	10/7/2021				x	x
13								
14		Formula in C8:						
15		=XLOOKUP("x",D8:H8,\$D\$7:\$H\$7)						
16		Enter and copy to rows below.						

Lookup last when there are duplicates: lookup last sale for employee

	A	B	C	D	E	F	G	H	I	J	K	L
1												
2		Goals: 1) Lookup last sale for each Sales Rep.										
3		"Get Last" when there are duplicates and date column is sorted A-Z.										
4												
5		Date	Sales Rep	Sales		Sales Rep	Last Sale	Formula in G6:				
6		10/3/2021	Sioux	\$640.56		Sioux	696.97	=XLOOKUP(F6:F7,C6:C15,D6:D15,,,-1)				
7		10/3/2021	Chin	\$1,706.02		Chin	765.99	Enter and it spills down to rows below.				
8		10/4/2021	Sioux	\$1,589.69								
9		10/4/2021	Chin	\$695.31								
10		10/5/2021	Sioux	\$474.26				Searching for the position last implies that the table has been sorted				
11		10/5/2021	Chin	\$786.13								
12		10/6/2021	Sioux	\$811.12								
13		10/6/2021	Chin	\$983.25								
14		10/7/2021	Sioux	\$696.97								
15		10/7/2021	Chin	\$765.99								

Two lookup values: lookup product price for a specified city

	A	B	C	D	E	F
1						
2		Goal: Lookup price based on product & store.				
3		Two value lookup to determine price.				
4						
5		Store	Seattle			
6		Product	Quad			
7		Price	39.95			
8						
9		Formula in C7:				
10		=XLOOKUP(C5&C6,B13:B21&C13:C21,D13:D21)				
11						
12		Store	Product	Price		
13		Oakland	Quad	43.95		
14		Oakland	Aspen	27.95		
15		Oakland	Carlota	25.95		
16		Seattle	Quad	39.95		
17		Seattle	Aspen	24.95		
18		Seattle	Carlota	26.95		
19		Tacoma	Quad	30.95		
20		Tacoma	Aspen	27.95		
21		Tacoma	Carlota	25.95		

Two-Way lookup: lookup tax based on weekly pay and allowances


	A	B	C	D	E	F	G	H	I	J	K	L
1												
2		Goal: Lookup federal income tax deduction based on gross pay and allowances.										
3		Perform 2-way look, with row header & column header determining intersecting lookup value.										
4												
5		SINGLE Person—WEEKLY Payroll Period										
6		If the gross pay is:		And the number of withholding allowances claimed is—								
7		At least	But less than	0	1	2	3	4	5	6	7	
8		The amount of income tax to be withheld is—										
9	Hidden rows	450	460	31	20	12	4	0	0	0	0	
10		460	470	33	22	13	5	0	0	0	0	
11		470	480	34	23	14	6	0	0	0	0	
32		680	690	66	55	44	33	22	13	0	0	
33		690	700	67	56	45	34	23	14	0	0	
34		700	710	68	57	45	34	23	14	0	0	
35		710	720	69	58	46	35	23	15	0	0	
36												
37		Gross Pay		705								
38		Allowance		2								
39		Fed Tax Deduction		45								
40												
41		Formula in D39:										
42		=XLOOKUP(D37,B9:B35,XLOOKUP(D38,D7:K7,D9:K35),"Not Found",-1)										
43												
44												
45		Outer XLOOKUP to lookup "700-710" row in "2" column.			Inner XLOOKUP to lookup all rows in "2" column. Placed in return_array of Outer XLOOKUP							
46												
47												
48												
49												
50												
51												
52				45								
53												
54		Formula in B52:										
55		=XLOOKUP(D38,D7:K7,XLOOKUP(D37,B9:B35,D9:K35,"Not Found",-1))										

FILTER array function arguments

Note: In Excel Basics video #2, we called a formula that makes a logical test on a column a “Array Logical Test”. In this section of the notes, I am referring to an “Array Logical Test” as “Boolean Array Logical Test” in reference to the great 19th century mathematician George Boole.

Note: For filtering data sets in place, you can use the filter feature (Data tab, Sort & Filter group). In addition, if you want to complete a quick filter on a data set and then copy and paste the records to a new location, the filter feature is easier to use than the FILTER function. However, the results from the filter feature do not instantly update when source data changes. When you use the FILTER function, or any worksheet formula, when the source data changes, the results from the FILTER function update instantly. This is the main advantage of the FILTER function over the filter feature.

The FILTER array function filters a dataset based on conditions and criteria that you specify in a Boolean Array Logical Test to deliver a resultant array of answers. The FILTER array function has the following syntax:



The arguments of the FILTER function are as follows:

array: This argument contains the table, column, or row that you want to filter.

include: This argument contains a Boolean Array Logical Test that indicates which items to include and which items to not include in the resultant array of answers. This array must have the same number of rows or columns as the array that is being filtered.

[if_empty]: This argument contains the value you want the formula to return when the result of the filter is an empty filter (that is, with no records). Here are two examples.

- Single-cell message: =FILTER(B5:D11,(B5:B11=F5)*(C5:C11=F8),"NA")
- Three-column message: =FILTER(B5:D11,(B5:B11=F5)*(C5:C11=F8),{"None","-","-"})

A Boolean Array Logical Test must evaluate to an array of TRUE, FALSE, or number values. If there are other values in the logical test resultant array, such as text values or errors, the FILTER function delivers a #VALUE! error. A TRUE value or any nonzero number means that the item should be included in the filtered array of answers. A FALSE value or zero means that the item should not be included in the filtered array of answers. Consider these examples (which can be found on the worksheet 3.4):

=FILTER({"Q";"C";"Q";"A"},{TRUE;FALSE;TRUE;FALSE}) = {"Q";"Q"}

=FILTER({"Q";"C";"Q";"A"},{1;0;1;0}) = {"Q";"Q"}

=FILTER({"Q";"C";"Q";"A"},{43;0;-2;0}) = {"Q";"Q"}

=FILTER({"Q";"C";"Q";"A"},{TRUE;0;-2;0}) = {"Q";"Q"}

=FILTER({"Q";"C";"Q";"A"},{TRUE;"";-2;"Text"}) = #VALUE!

If you are filtering by rows, a Boolean Array Logical Test must contain a single column that contains the same number of rows as there are in the item being filtered in the *array* argument. If the numbers of rows in the *array* and *include* arguments are not equivalent, the result is a #VALUE! error. Consider these examples (which can be found on the worksheet 3.4):

=FILTER({"Q";"C";"Q";"A"},{TRUE;FALSE;TRUE;FALSE}) = {"Q";"Q"}

=FILTER({"Q";"C";"Q";"A"},{TRUE;FALSE;TRUE}) = #VALUE!

If you are filtering by columns, a Boolean Array Logical Test must contain a single row that contains the same number of columns as there are in the item being filtered in the *array* argument. If the number of columns in the *array* and *include* arguments are not equivalent, the result is a #VALUE! error. Consider these examples (which can be found on the worksheet 3.4):

=FILTER({"Q";"C";"Q";"A"},{TRUE,FALSE,TRUE,FALSE}) = {"Q";"Q"}

=FILTER({"Q";"C";"Q";"A"},{TRUE,FALSE,TRUE}) = #VALUE!

The include argument in the FILTER function can contain a Single-Condition Logical Test, an AND Logical Test, an OR Logical Test, a NOT Logical Test, and many other types of logical tests. Here is a list of examples of each of these different logical tests:

Single-condition logical test: =FILTER(B5:D11,B5:B11=F5)

AND logical test using multiplication: =FILTER(B5:D11,(B5:B11=F5)*(C5:C11=F8))

OR logical test using addition: =FILTER(B5:D11,(B5:B11=F5)+(C5:C11=F8))


NOT logical test: =FILTER(B5:D11,B5:B11<>F5)

Use an AND and OR logical test together (complex filter): =FILTER(B11:B16,(C11:C16>C7)*(D11:D16>D7)*(E11:E16<E7)*((F11:F16>F7)+(G11:G16>G7)))

FILTER function to return multiple items from one lookup value

	A	B	C	D	E	F	G	H	I	J	K	L	M
1													
2		Goal: Extract student class records.											
3		Famous Excel task: One lookup value, return multiple items/records.											
4													
5		Student	Class	Department	Grade		Student						
6		Sioux Chin	Acc 121	Accounting	1.7		Gigi Dmitri						
7		Chantel Mimms	Busn 216	Business	3.1								
8		Dylan Franks	Busn 101	Business	2.1		Class	Department	Grade	Formula in G9:			
9		Sioux Chin	Acc 201	Accounting	4		Busn 210	Business	2.6	=FILTER(C6:E17,B6:B17=G6)			
10		Chantel Mimms	Busn 218	Business	3.8		Econ 202	Economics	2				
11		Dylan Franks	Eng 201	English	2								
12		Sioux Chin	Econ 201	Economics	2.6								
13		Chantel Mimms	Busn 210	Business	3.3								
14		Dylan Franks	Busn 216	Business	1.9		Class	Department	Grade				
15		Gigi Dmitri	Busn 210	Business	2.6		Busn 210	Business	2.6				
16		Chantel Mimms	Eng 201	English	3.1		Econ 202	Economics	2				
17		Gigi Dmitri	Econ 202	Economics	2								
18													
19							Formula 'before we had FILTER' in G15:						
20							=IF(ROWS(G\$15:G15)>COUNTIFS(\$B\$6:\$B\$17,\$G\$6),"",						
21							INDEX(C\$6:C\$17,AGGREGATE(15,6,(ROW(\$B\$6:\$B\$17)-						
22							ROW(\$B\$6)+1)/(\$B\$6:\$B\$17=\$G\$6),ROWS(G\$15:G15)))						
23													

XMATCH function arguments



XMATCH lookup function to compare two lists


	A	B	C	D	E	F	G	H	I	J	K
1											
2	Goal: Are names in Sunday list also in the Saturday list?										
4	Saturday List		Sunday List		Are Items in Both Lists?						
6	Signups Sat. Event		Signups Sun. Event		In Both?		Formula in E7:				
7	Ty		Miki		FALSE		=ISNUMBER(XMATCH(D7:D11,B7:B10))				
8	Gigi		Gigi		TRUE						
9	Sioux		Kip		FALSE						
10	Chantel		Chantel		TRUE						
11			Fran		FALSE						
12											
13	Goal: Are names in Sunday list NOT in the Saturday list?										
14											
15	Saturday List		Sunday List		Are Items From This List In The Other List?						
17	Signups Sat. Event		Signups Sun. Event		In Both?		Formula in E18:				
18	Ty		Miki		TRUE		=ISNA(XMATCH(D18:D22,B18:B21))				
19	Gigi		Gigi		FALSE						
20	Sioux		Kip		TRUE						
21	Chantel		Chantel		FALSE						
22			Fran		TRUE						

SWITCH function arguments

The SWITCH lookup function is the Swiss Army knife of the lookup function family because it is not limited to values in cells, ranges, or arrays (as the XLOOKUP and LOOKUP functions are). With the SWITCH function, you can look up tables, formulas, functions, text values, number values, values in cells, and any other type of formula element that you can dream up. Because it is a function that can look up any type of formula element or value, SWITCH is also an efficient substitute for the IF or IFS functions in some situations.

The arguments of the SWITCH function are as follows:

- expression: This argument specifies the lookup value.
- value1, value2, etc.: These arguments specify the values to match.
- result1, result2, etc.: These arguments specify the values returned.
- default: This argument is delivered when no matches are found in the value arguments.



SWITCH to lookup the correct pricing lookup table based on product name

The Figure above shows a close-up of the SWITCH function used in next example, where you have to look up different lookup tables based on a product name. The relative cell reference B6 contains the name of the product sold. If the product in cell B6 is Bellen, then the table in the range J6:K8 is used; if the product is Quad, the table in the range J11:K13 is used; finally, if the product is not Bellen or Quad, the default table from the range J16:K18 is used.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
2		Goal:													
3		1) With SWITCH function, lookup correct lookup table based on product name.													
4		2) Then use SWITCH result in the LOOKUP function to perform lookup for Unit "Exact Match or Next Smaller with Binary Search"													
6		Units Transaction Table:											IFS is longer formula		
8		Product	Price	Units Sold	Price Discount				Bellen	Units	P. Discount		Price Discount		
9		Bellen	26.95	24	0.25					0	0%		0.25		
10		Aspen	28.95	60	0.5					15	25%		0.5		
11		Bellen	26.95	50	0.4					45	40%		0.4		
12		Quad	43.95	5	0								0		
13		Quad	43.95	25	0.2			Quad	Units	P. Discount			0.2		
14		Bellen	26.95	96	0.4					0	0%		0.4		
15		Yanaki	30.95	8	0					20	20%		0		
16		Quad	43.95	124	0.45					60	45%		0.45		
17		Bellen	26.95	55	0.4								0.4		
18		Bellen	26.95	2	0			Default	Units	P. Discount			0		
19		Quad	43.95	5	0					0	0%		0		
20		Yanaki	30.95	19	0.3					15	30%		0.3		
21										25	50%				
22															
23		Formula in E9:=LOOKUP(D9,SWITCH(B9,\$G\$8,\$I\$9:\$J\$11,\$G\$13,\$I\$14:\$J\$16,\$I\$19:\$J\$21))													

LOOKUP function to create formula much more quickly for tax calculations

F21 =LOOKUP(F20,B6:F13)+(F20-LOOKUP(F20,B6:E13))*LOOKUP(F20,B6:D13)						
A	B	C	D	E	F	G
2	Ex	Goal: "Approximate match" lookup when 1st column sorted & we want item from last column.				
3	#23 Advantage:	Enter fewer arguments, as compared to XLOOKUP. Faster to create formula with LOOKUP.				
4						
5	Taxable Pay Lower Limit	Taxable Pay Upper Limit	Tax Rate	Upper Limit Previous Category	Cumulative Tax From Previous Categories	Tax Rule in Full:
6	\$0.00	\$222	0%	\$0	\$0.00	
7	\$222.01	\$588	10%	\$222	\$0.00	\$0.00 +(TP - \$222) *10%
8	\$588.01	\$1,711	12%	\$588	\$36.60	\$36.60 +(TP - \$588) *12%
9	\$1,711.01	\$3,395	22%	\$1,711	\$171.36	\$171.36 +(TP - \$1,711) *22%
10	\$3,395.01	\$6,280	24%	\$3,395	\$541.84	\$541.84 +(TP - \$3,395) *24%
11	\$6,280.01	\$7,914	32%	\$6,280	\$1,234.24	\$1,234.24 +(TP - \$6,280) *32%
12	\$7,914.01	\$11,761	35%	\$7,914	\$1,757.12	\$1,757.12 +(TP - \$7,914) *35%
13	\$11,761.01	more	37%	\$11,761	\$3,103.57	\$3,103.57 +(TP - \$11,761) *37%
14						
15	If we make calculation manually: $\$541.84 + (\$3,690.80 - \$3,395) * 24\% = \612.83					
16						
17	Gross Pay				\$4,010.00	
18	Withholding Allowance				\$79.80	
19	# of Withholding Allowances				4	
20	Taxable Pay (TP). This is lookup_value:				\$3,690.80	
21	Federal Income Tax Withholdings?				\$612.83	
22	Federal Income Tax Withholdings?				\$612.83	
23						
24	Formula in F21:=LOOKUP(F20,B6:F13)+(F20-LOOKUP(F20,B6:E13))*LOOKUP(F20,B6:D13)					
25	Formula in F22:=XLOOKUP(F20,B6:B13,F6:F13,,-1)+(F20-XLOOKUP(F20,B6:B13,E6:E13,,-1))*XLOOKUP(F20,B6:B13,D6:D13,,-1)					